



Trabajo Práctico N° 8

Alcance, Entornos, Pasaje de Parámetros

Ejercicio 1: Considere el siguiente programa en Lenguaje Pascal:

```
PROGRAM mayus;
CONST offset = ord('0');
VAR car: char;

FUNCTION EsMayuscula(c:char):boolean;
BEGIN {EsMayuscula}
  EsMayuscula := (c >= 'A') and (c <= 'Z')
END; {EsMayuscula}

FUNCTION EsMinuscula(car:char):boolean;
BEGIN {EsMinuscula}
  EsMinuscula := (car >= 'a') and (car <= 'z')
END; {EsMinuscula}

FUNCTION AMayuscula(c:char):char;
CONST offset = ord('A') - ord('a');
VAR car: char;
BEGIN {AMayuscula}
  IF esMayuscula(c) THEN
    car := c
  ELSE
    car := chr(ord(c) + offset);
  AMayuscula := car;
END; {AMayuscula}

FUNCTION EsVocal(c:char):boolean;
VAR minus:char;

  FUNCTION AMinuscula(c:char):char;
  CONST offset = ord('a') - ord('A');
  BEGIN {AMinuscula}
    IF esMayuscula(c) THEN
      minus := chr(ord(c) + offset)
    ELSE
      minus := c;
    AMinuscula := minus;
  END; {AMinuscula}

BEGIN {EsVocal}
  minus := AMinuscula(c);
  EsVocal := (minus = 'a') OR (minus = 'e') OR (minus = 'i') OR (minus = 'o') OR
(minus = 'u')
END; {EsVocal}

BEGIN {Ppal}
  writeln('Ingrese una frase terminada en punto "."');
  REPEAT
    read(car);
    IF EsMayuscula(car) OR EsMinuscula(car) THEN
      IF EsVocal(car) THEN write(AMayuscula(car))
      ELSE write(car);
    UNTIL (car = '\.');
END. {Ppal}
```



- a) Identifique los bloques del programa y los identificadores declarados en cada bloque.
- b) Complete la siguiente tabla indicando para cada bloque su entorno de referencia (**no** incluya aquellos identificadores que quedan *ocultos*).

	mayus	EsMayuscula	EsMinuscula	AMayuscula	EsVocal	AMinuscula
Entorno Global						
Entorno No Local						
Entorno Local						

- c) Como se afectaría la tabla del inciso anterior si se efectuaran sobre el programa anterior los siguientes cambios (considere los cambios en forma exclusiva, es decir, cuando considere algún inciso no efectúe ningún otro cambio de los otros incisos)
 - a. Si se agregara una variable AMinuscula de tipo char al bloque mayus (bloque del programa ppal).
 - b. Si se cambiara la función AMayuscula por la siguiente (observe que se deberá modificar la llamada):

```

FUNCTION AMayuscula:char;
CONST offset = ord('A') - ord('a');
BEGIN {AMayuscula}
    IF esMayuscula(car) THEN
        AMayuscula := car
    ELSE
        AMayuscula := chr(ord(car) + offset)
END; {AMayuscula}
    
```

- c. Si se cambiara la función AMinuscula por la siguiente:

```

FUNCTION AMinuscula(c:char):char;
FUNCTION offset(car:char):integer;
BEGIN
    offset := ord(car) + ord('a') - ord('A');
END;
BEGIN {AMinuscula}
    IF esMayuscula(c) THEN
        minus := chr(offset(c))
    ELSE
        minus := c;
    AMinuscula := minus;
END; {AMinuscula}
    
```



Ejercicio 2: Dado el siguiente programa en Pascal, identifique los bloques y para cada bloque su entorno de referencia (no incluya aquellos identificadores que quedan ocultos).

```
PROGRAM Entorno;
CONST Z=100;
VAR A, B:INTEGER;          EsPar:BOOLEAN;

PROCEDURE Multiplica(VAR X:INTEGER; Y:INTEGER);forward;

  PROCEDURE Cubo(VAR A:INTEGER);
  BEGIN
    A:= A * A * A
  END;

  PROCEDURE Multiplica(VAR X:INTEGER; Y:INTEGER);
  VAR Z:INTEGER;
  FUNCTION EsPar(A:INTEGER):BOOLEAN;forward;

    FUNCTION EsMayor(A,B:INTEGER):BOOLEAN;
    BEGIN
      EsMayor:= A > B
    END;

    FUNCTION EsPar(A:INTEGER):BOOLEAN;
    BEGIN
      EsPar := (A mod 2 = 0)
    END;

  BEGIN
    X:= X + 10;
    IF EsMayor(X,Y) THEN Cubo(Y)
  END;

BEGIN
  A:= 2; B:= 6;
  Multiplica(A, B);
  WRITELN(A, ' - ', B);
  Cubo(B);
  WRITELN(A, ' - ', B);    readln;

END.
```

a) Realice la traza del programa ¿Qué se muestra por pantalla como resultado de la ejecución del programa Entorno?

- | | | | | |
|--------------------|---------------------|-----------------------|----------------------------|------------------------------|
| 1) 2 - 6
2 - 6 | 3) 2 - 6
216 - 6 | 5) 12 - 6
12 - 216 | 7) 2 - 216
2 - 216 | 9) 2 - 216
10077696 - 216 |
| 2) 2 - 6
2 - 18 | 4) 12 - 6
12 - 6 | 6) 12 - 6
216 - 6 | 8) 2 - 216
2 - 10077696 | 10) 2 - 18
8 - 18 |

b) ¿Cuál es el alcance del identificador A definido en el ambiente Entorno?

- | | | | |
|--|---------------------|---------------------------|--|
| 1) Entorno,
Multiplica, Cubo,
EsPar, EsMayor | 2) Entorno,
Cubo | 3) Entorno,
Multiplica | 4) Entorno,
Multiplica,
EsPar, EsMayor |
|--|---------------------|---------------------------|--|

c) ¿En qué ambientes de referenciamiento es visible el identificador Z definido en el ambiente Multiplica?

- | | | | |
|--|---------|----------------------------------|---------------|
| 1) Entorno,
Multiplica, Cubo,
EsPar, EsMayor | 2) Cubo | 3) Multiplica,
EsPar, EsMayor | 4) Multiplica |
|--|---------|----------------------------------|---------------|

¿El identificador EsPar definido en el ambiente Entorno es visible en el ambiente Multiplica? Justifique su respuesta.